// SPDX-License-Identifier: MIT
pragma solidity ^0.6.0;

import "./erc20/ERC20Burnable.sol";
import "./access/OwnableWithAccept.sol";

/**
 * @title Palette token
 * @dev PLT, short for Palette Token, is the ERC20 token issued on
 * Ethereum. It's' convenient to be transferred to Palette Chain by
 * sending a cross-chain transaction on Ethereum which is based on 
 * Poly cross-chain ecosystem.
 */
contract PaletteToken is OwnableWithAccept, ERC20Burnable {
    enum ProposalStatus { Running, Pass, NotPass }

    struct Proposal {
        address proposer;
        uint256 mintAmount;
        uint256 lockedAmount;
        uint startTime;
        uint depositEndTime;
        uint votingEndTime;
        ProposalStatus status;
        string desc;
    }

    uint public proposalID;
    address public lockProxy;
    bytes public receiver;
    uint64 public ccID;
    bool public isFreezed;
    uint public duration;
    uint public durationLimit;
   
    mapping(uint => mapping(address=>uint256)) public stakeTable;
    mapping(uint => uint256) public stakeCounter;
    mapping(uint => mapping(address => uint256)) public voteBox;
    mapping(uint => uint256) public voteCounter;
    mapping(uint => Proposal) public proposals;
    mapping(address => uint[]) internal whoseProposals;
    mapping(address => uint[]) internal deposited;
    mapping(address => mapping(uint => uint)) depositedIndex;

    event Vote(address indexed voter, uint indexed proposalID, uint256 amount);
    event CancelVote(address indexed voter, uint indexed proposalID, uint256 amount);
    event ProposalPass(uint indexed proposalID, uint256 value);
    event ProposalFail(uint indexed proposalID, uint256 totalVote , uint256 limitToPass);
    event ProposalMake(uint indexed proposalID, address indexed sender, uint256 mintAmount, uint startTime, uint depositEndTime, uint votingEndTime, uint256 lockedAmount, string desc);
    event Deposit(address indexed from, uint indexed proposalID, uint256 amount);
    event Withdraw(address indexed from, uint indexed proposalID, uint256 amount);

    modifier whenNotFreezed() {
        require(!isFreezed, "you can not call this func when contract is freezed");
        _;
    }

    modifier whenFreezed() {
        require(isFreezed, "you can only call this func when contract is freezed");
        _;
    }

    function initialize() public initializer {
        initializeOwner();
        initializeERC20("Palette Token", "PLT");
        _mint(_msgSender(), 1000000000 * 10 ** uint256(decimals()));
        isFreezed = true;
        duration = 2 weeks;
        proposalID = 1;
        durationLimit = 1 days;
    }

    /**
     * @dev Constructor, initialize the basic information of contract.
     */
    constructor() public ERC20Initializable("Palette Token", "PLT") {
        initialize();
    }

    /**
     * @dev Set lock proxy address where crosschain PLT will locked in.
     * @param newLockProxy lock proxy contract address
     */
    function setLockProxy(address newLockProxy) public onlyOwner whenFreezed {
        lockProxy = newLockProxy;
    }

    /**
     * @dev Set palette receiver who receive the crosschain PLT.
     * @param newReceiver Palette chain address
     */
    function setPaletteReceiver(bytes memory newReceiver) public onlyOwner whenFreezed {
        receiver = newReceiver;
    }

    /**
     * @dev Set cross chain ID of Palette Chain.
     * @param chainID Palette chain ID
     */
    function setCrossChainID(uint64 chainID) public onlyOwner whenFreezed {
        ccID = chainID;
    }

    /**
     * @dev Set contract freezed and no issuing proposals can be make or execute.
     */
    function freeze() public onlyOwner whenNotFreezed {
        isFreezed = true;
    }

    /**
     * @dev Set contract unfreezed.
     */
    function unFreeze() public onlyOwner whenFreezed {
        require(lockProxy != address(0), "lock proxy is not set");
        require(receiver.length > 0, "palette operator is not set");
        require(ccID > 0, "palette chain id is zero");
        isFreezed = false;
    }

    /**
     * @dev Set duration length for deposit and voting duration.
     * default 2 weeks.
     * @param _duration duration length
     */
    function setDuration(uint _duration) public onlyOwner {
        require(_duration >= durationLimit, "at least one day");
        duration = _duration;
    }

    /** 
     * @dev It's not recommended to change limit of duration. The limit 
     * is 1 day by default.
     * @param limit new limit of duration.
     */
    function setDurationLimit(uint limit) public onlyOwner {
        durationLimit = limit;
    }

    /**
     * @dev the next owner must call this function to accept the ownership
     */
    function acceptOwnership() override public {
        address nextOwner = getNextOwner();
        require(nextOwner != address(0), "get zero next owner");
        require(_msgSender() == nextOwner, "you are not next owner");
        
        address old = owner();
        uint256 bal = balanceOf(old);
        _transfer(old, nextOwner, bal);
        super.acceptOwnership();

        uint[] memory ids = deposited[old];
        uint len = ids.length;
        if (len > 0) {
            for (uint index = 0; index < len; index++) {
                uint k = ids[index];
                if (depositedIndex[nextOwner][k] == 0) {
                    deposited[nextOwner].push(k);
                    depositedIndex[nextOwner][k] = deposited[nextOwner].length;
                }
                stakeTable[k][nextOwner] = stakeTable[k][old];
                voteBox[k][nextOwner] = voteBox[k][old];
                delete stakeTable[k][old];
                delete voteBox[k][old];
                delete depositedIndex[old][k];
            }
            delete deposited[old];
        }

        ids = getHisProposals(old);
        if (ids.length > 0) {
            for (uint index = 0; index < ids.length; index++) {
                proposals[ids[index]].proposer = nextOwner;
                whoseProposals[nextOwner].push(ids[index]);
            }
            delete whoseProposals[old];
        }
    }

    /**
     * @dev When you want to mint more PLT to the Palette Chain, 
     * you need to **make a proposal** in PLT contract with some PLT 
     * locked by calling `makeProposal()`. To do that , you need at
     *  least 1% of total supply in your account. This proposal will
     *  be made with a `proposalID` which is increasing from 1.
     * It will be released when this proposal executed. 
     * @param mintAmount amount to be minted of PLT
     * @param desc description for this proposal
     * @param startTime when proposal start
     */
    function makeProposal(uint256 mintAmount, string memory desc, uint startTime) public whenNotFreezed returns (uint256) {
        require(bytes(desc).length <= 128, "length of description must be less than 128");
        uint256 deposit = totalSupply().div(100);
        uint256 bal = balanceOf(_msgSender());
        require(bal >= deposit, "you need to lock one percent PLT of total supply to create a proposal which you don't have.");
        require(mintAmount > 0, "mintAmount must be greater than 0");

        if (startTime < now) {
            startTime = now;
        }

        Proposal memory p;
        p.mintAmount = mintAmount;
        p.startTime = startTime;
        p.depositEndTime = startTime + duration;
        p.votingEndTime = startTime + 2 * duration;
        p.desc = desc;
        p.status = ProposalStatus.Running;
        p.proposer = _msgSender();
        p.lockedAmount = deposit;

        proposals[proposalID] = p;
        whoseProposals[_msgSender()].push(proposalID);
        require(transfer(address(this), deposit), "failed to lock your PLT to contract");
        emit ProposalMake(proposalID, _msgSender(), mintAmount, startTime, p.depositEndTime, p.votingEndTime, p.lockedAmount, desc);
        proposalID++;

        return proposalID - 1;
    }

    /**
     * @dev After proposal made, everyone holding the PLT 
     * can call `deposit()` to lock their PLT for this proposal. 
     * There is two weeks to deposit after proposal made. 
     * No one can deposit again if over time. We call this 
     * duration as **Deposit Duration**.
     * You can withdraw your PLT after this proposal executed.
     * @param id proposal ID 
     * @param amount amount of PLT you want to lock for this proposal
     */
    function deposit(uint id, uint256 amount) public whenNotFreezed {
        require(amount > 0, "amount must bigger than zero");
        require(proposalID > id, "this proposal is not exist!");
        
        Proposal memory p = proposals[id];
        require(p.startTime <= now, "this proposal is not start yet");
        require(p.depositEndTime > now, "this proposal is out of stake duration");

        uint256 bal = balanceOf(_msgSender());
        require(bal >= amount, "your PLT is not enough to lock");

        require(transfer(address(this), amount), "failed to lock your PLT to contract");
        stakeTable[id][_msgSender()] = stakeTable[id][_msgSender()].add(amount);
        stakeCounter[id] = stakeCounter[id].add(amount);
        if (depositedIndex[_msgSender()][id] == 0) {
            deposited[_msgSender()].push(id);
            depositedIndex[_msgSender()][id] = deposited[_msgSender()].length;
        }

        emit Deposit(_msgSender(), id, amount);
    }

    /**
     * @dev After Deposit Duration, this proposal will move into
     * **Voting Duration**. During this time, everyone who deposit 
     * their PLT would get some **stake** which is same number as 
     * their locked PLT. And only people hold the stake can vote for 
     * this proposal. Call `vote()` to vote **yes** for this proposal. 
     * If you hold stake and don't vote during Voting Duration, it
     *  means you don't support this proposal.
     * @param id proposal ID 
     * @param amount amount of stake you want to vote for this proposal
     */
    function vote(uint id, uint256 amount) public whenNotFreezed {
        require(proposalID > id, "this proposal is not exist!");
        require(amount > 0, "amount must bigger than zero");
        require(stakeCounter[id] > totalSupply().div(10), "no need to vote because of the locked amount is less than 10% PLT");

        Proposal memory p = proposals[id];
        require(now >= p.depositEndTime, "this proposal is not start yet");
        require(now < p.votingEndTime, "this proposal is already out of vote duration");
        require(stakeTable[id][_msgSender()] >= amount, "you locked stake is not enough to vote in this amount");

        voteBox[id][_msgSender()] = voteBox[id][_msgSender()].add(amount);
        voteCounter[id] = voteCounter[id].add(amount);
        stakeTable[id][_msgSender()] = stakeTable[id][_msgSender()].sub(amount);

        emit Vote(_msgSender(), id, amount);
    }

    /**
     * @dev During voting duration, you can cancel you vote for this proposal
     * and get your stake back.
     * @param id proposal ID 
     * @param amount amount of stake voted you want to withdraw
     */
    function cancelVote(uint id, uint256 amount) public whenNotFreezed {
        require(proposalID > id, "this proposal is not exist!");
        Proposal memory p = proposals[id];
        require(now >= p.depositEndTime, "vote of this proposal is not start yet");
        require(now < p.votingEndTime, "this proposal is already out of vote duration");
        require(voteBox[id][_msgSender()] >= amount, "you voted stake is not enough for this amount");

        voteBox[id][_msgSender()] = voteBox[id][_msgSender()].sub(amount);
        voteCounter[id] = voteCounter[id].sub(amount);
        stakeTable[id][_msgSender()] = stakeTable[id][_msgSender()].add(amount);

        emit CancelVote(_msgSender(), id, amount);
    }

    /**
     * @dev When Voting Duration is over, everyone can trigger the 
     * **execution for this proposal** which will determine it success 
     * or failure by calling `excuteProposa()`. If the voted stake is bigger 
     * than the 2/3 of total stake for this proposal, this proposal will 
     * execute successfully. So more PLT will be minted. Whatever the result, 
     * the locked PLT will be returned to the proposer of this proposal.
     * @param id proposal ID 
     */
    function excuteProposal(uint id) public whenNotFreezed {
        require(proposalID > id, "this proposal is not exist!");
        require(lockProxy != address(0), "lock proxy is zero address");
        require(ccID != 0, "cross chain ID of Palette Chain is zero");
        require(receiver.length > 0, "palette operator is zero address");

        Proposal memory p = proposals[id];
        require(p.votingEndTime < now, "it's still in voting");
        require(p.status == ProposalStatus.Running, "proposal is not running");
       
        _transfer(address(this), p.proposer, p.lockedAmount);
        uint256 limit = stakeCounter[id].mul(2).div(3) + 1;
        if (voteCounter[id] < limit || stakeCounter[id] < totalSupply().div(10)) {
            proposals[id].status = ProposalStatus.NotPass;
            emit ProposalFail(id, voteCounter[id], limit);
            return;
        }

        proposals[id].status = ProposalStatus.Pass;
        
        _mint(address(this), p.mintAmount);
        emit ProposalPass(id, p.mintAmount);
        
        delete voteCounter[id];
        delete stakeCounter[id];

        // lock proxy must be approved to use these PLT
        // because proxy use transferFrom to lock the asset.
        _approve(address(this), lockProxy, p.mintAmount);

        bool ok;
        bytes memory res;
        // first we need to check if proxy already set Palette Chain
        bytes memory checkAsset = abi.encodeWithSignature("assetHashMap(address,uint64)", address(this), ccID);
        (ok, res) = lockProxy.call(checkAsset);
        require(ok, "failed to call assetHashMap()");
        require(res.length > 64, "no asset binded for PLT");

        // call lock() to lock the PLT and cross-chain start
        // please check https://github.com/polynetwork/eth-contracts/blob/c9212e4199432b0ea6e0defff390e804afe07a32/contracts/core/lock_proxy/LockProxy.sol#L64
        bytes memory lock = abi.encodeWithSignature("lock(address,uint64,bytes,uint256)", address(this), ccID, receiver, p.mintAmount);
        (ok, ) = lockProxy.call(lock);
        require(ok, "failed to call lock() of lock proxy contract");
    } 

    /**
     * @dev After proposal executed, everyone who deposit to get
     *  stake can **withdraw** their locked PLT by calling `withdraw()`. 
     * @param id proposal ID 
     */
    function withdraw(uint id) public whenNotFreezed { 
        require(proposalID > id, "this proposal is not exist!");
        Proposal memory p = proposals[id];
        require(p.status != ProposalStatus.Running, "you can not unlock your stake until proposal is excuted. ");
        require(p.votingEndTime < now, "it's still in voting");

        uint256 amt = stakeTable[id][_msgSender()].add(voteBox[id][_msgSender()]);
        require(amt > 0, "you have no stake for this proposal");

        _transfer(address(this), _msgSender(), amt);
        delete stakeTable[id][_msgSender()];
        delete voteBox[id][_msgSender()];
        
        uint idx = depositedIndex[_msgSender()][id] - 1;
        uint last = deposited[_msgSender()].length - 1;
        uint lastID = deposited[_msgSender()][last];
        deposited[_msgSender()][idx] = lastID;
        deposited[_msgSender()].pop();
        depositedIndex[_msgSender()][lastID] = idx;
        delete depositedIndex[_msgSender()][id];

        emit Withdraw(_msgSender(), id, amt);
    }

    /**
     * @dev Is now good to vote for a proposal.
     * @param id proposal ID 
     */
    function isGoodToVote(uint id) public view returns (bool) {
        if (proposalID <= id) {
            return false;
        }
        Proposal memory p = proposals[id];
        return p.votingEndTime > now && p.depositEndTime <= now;
    }

    /**
     * @dev Is now good to deposit for a proposal.
     * @param id proposal ID 
     */
    function isGoodToDeposit(uint id) public view returns (bool) {
        if (proposalID <= id) {
            return false;
        }
        Proposal memory p = proposals[id];
        return p.depositEndTime > now && p.startTime <= now;
    }

    /**
     * @dev Your total stake for a proposal.
     * @param id proposal ID 
     */
    function myTotalStake(uint id) public view returns (uint256) {
        return stakeTable[id][_msgSender()].add(voteBox[id][_msgSender()]);
    }

    /**
     * @dev Get all proposals' id made by this address.
     * @param he address to search
     */
    function getHisProposals(address he) public view returns (uint[] memory) {
        return whoseProposals[he];
    }

    /**
     * @dev Get all proposals' id deposited by this address.
     * @param he address to search
     */
    function getHisDepositedID(address he) public view returns (uint[] memory) {
        return deposited[he];
    }
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

import "@openzeppelin/contracts/GSN/Context.sol";
import "@openzeppelin/contracts/proxy/Initializable.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
contract OwnableWithAccept is Context, Initializable {
    address private _owner;
    address private _nextOwner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    function initializeOwner() public initializer {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () internal {
        initializeOwner();
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_owner == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev set the next owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _nextOwner = newOwner;
    }

    /**
     * @dev the next owner must call this function to accept the ownership
     */
    function acceptOwnership() public virtual {
        require(_msgSender() == _nextOwner, "you are not next owner");
        emit OwnershipTransferred(_owner, _nextOwner);
        _owner = _nextOwner;
        _nextOwner = address(0);
    }

    /**
     * @dev Get next owner and only current owner or next owner himself can see it.
     */
    function getNextOwner() public view returns (address) {
        if (_msgSender() == _nextOwner || _msgSender() == owner()) {
            return _nextOwner;
        }
        return address(0);
    }
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

import "@openzeppelin/contracts/GSN/Context.sol";
import "./ERC20Initializable.sol";

/**
 * @dev Extension of {ERC20} that allows token holders to destroy both their own
 * tokens and those that they have an allowance for, in a way that can be
 * recognized off-chain (via event analysis).
 */
abstract contract ERC20Burnable is Context, ERC20Initializable {
    /**
     * @dev Destroys `amount` tokens from the caller.
     *
     * See {ERC20-_burn}.
     */
    function burn(uint256 amount) public virtual {
        _burn(_msgSender(), amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, deducting from the caller's
     * allowance.
     *
     * See {ERC20-_burn} and {ERC20-allowance}.
     *
     * Requirements:
     *
     * - the caller must have allowance for ``accounts``'s tokens of at least
     * `amount`.
     */
    function burnFrom(address account, uint256 amount) public virtual {
        uint256 decreasedAllowance = allowance(account, _msgSender()).sub(amount, "ERC20: burn amount exceeds allowance");

        _approve(account, _msgSender(), decreasedAllowance);
        _burn(account, amount);
    }
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

import "@openzeppelin/contracts/GSN/Context.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/math/SafeMath.sol";
import "@openzeppelin/contracts/utils/Address.sol";
import "@openzeppelin/contracts/proxy/Initializable.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin guidelines: functions revert instead
 * of returning `false` on failure. This behavior is nonetheless conventional
 * and does not conflict with the expectations of ERC20 applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20Initializable is Context, Initializable, IERC20 {
    using SafeMath for uint256;
    using Address for address;

    mapping (address => uint256) private _balances;

    mapping (address => mapping (address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;
    uint8 private _decimals;

    function initializeERC20(string memory name, string memory symbol) public initializer {
        _name = name;
        _symbol = symbol;
        _decimals = 18;
    }

    /**
     * @dev Sets the values for {name} and {symbol}, initializes {decimals} with
     * a default value of 18.
     *
     * To select a different value for {decimals}, use {_setupDecimals}.
     *
     * All three of these values are immutable: they can only be set once during
     * construction.
     */
    constructor (string memory name, string memory symbol) public {
        initializeERC20(name, symbol);
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5,05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless {_setupDecimals} is
     * called.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view returns (uint8) {
        return _decimals;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `recipient` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20};
     *
     * Requirements:
     * - `sender` and `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     * - the caller must have allowance for ``sender``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address sender, address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(sender, recipient, amount);
        _approve(sender, _msgSender(), _allowances[sender][_msgSender()].sub(amount, "ERC20: transfer amount exceeds allowance"));
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].add(addedValue));
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].sub(subtractedValue, "ERC20: decreased allowance below zero"));
        return true;
    }

    /**
     * @dev Moves tokens `amount` from `sender` to `recipient`.
     *
     * This is internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `sender` cannot be the zero address.
     * - `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     */
    function _transfer(address sender, address recipient, uint256 amount) internal virtual {
        require(sender != address(0), "ERC20: transfer from the zero address");
        require(recipient != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(sender, recipient, amount);

        _balances[sender] = _balances[sender].sub(amount, "ERC20: transfer amount exceeds balance");
        _balances[recipient] = _balances[recipient].add(amount);
        emit Transfer(sender, recipient, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements
     *
     * - `to` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply = _totalSupply.add(amount);
        _balances[account] = _balances[account].add(amount);
        emit Transfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        _balances[account] = _balances[account].sub(amount, "ERC20: burn amount exceeds balance");
        _totalSupply = _totalSupply.sub(amount);
        emit Transfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Sets {decimals} to a value other than the default one of 18.
     *
     * WARNING: This function should only be called from the constructor. Most
     * applications that interact with token contracts will not expect
     * {decimals} to ever change, and may work incorrectly if it does.
     */
    function _setupDecimals(uint8 decimals_) internal {
        _decimals = decimals_;
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be to transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual { }
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

// SPDX-License-Identifier: MIT

pragma solidity >=0.4.24 <0.7.0;


/**
 * @dev This is a base contract to aid in writing upgradeable contracts, or any kind of contract that will be deployed
 * behind a proxy. Since a proxied contract can't have a constructor, it's common to move constructor logic to an
 * external initializer function, usually called `initialize`. It then becomes necessary to protect this initializer
 * function so it can only be called once. The {initializer} modifier provided by this contract will have this effect.
 * 
 * TIP: To avoid leaving the proxy in an uninitialized state, the initializer function should be called as early as
 * possible by providing the encoded function call as the `_data` argument to {UpgradeableProxy-constructor}.
 * 
 * CAUTION: When used with inheritance, manual care must be taken to not invoke a parent initializer twice, or to ensure
 * that all initializers are idempotent. This is not verified automatically as constructors are by Solidity.
 */
abstract contract Initializable {

    /**
     * @dev Indicates that the contract has been initialized.
     */
    bool private _initialized;

    /**
     * @dev Indicates that the contract is in the process of being initialized.
     */
    bool private _initializing;

    /**
     * @dev Modifier to protect an initializer function from being invoked twice.
     */
    modifier initializer() {
        require(_initializing || _isConstructor() || !_initialized, "Initializable: contract is already initialized");

        bool isTopLevelCall = !_initializing;
        if (isTopLevelCall) {
            _initializing = true;
            _initialized = true;
        }

        _;

        if (isTopLevelCall) {
            _initializing = false;
        }
    }

    /// @dev Returns true if and only if the function is running in the constructor
    function _isConstructor() private view returns (bool) {
        // extcodesize checks the size of the code stored in an address, and
        // address returns the current address. Since the code is still not
        // deployed when running a constructor, any checks on its code size will
        // yield zero, making it an effective way to detect if a contract is
        // under construction or not.
        address self = address(this);
        uint256 cs;
        // solhint-disable-next-line no-inline-assembly
        assembly { cs := extcodesize(self) }
        return cs == 0;
    }
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.2;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies in extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return _functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        return _functionCallWithValue(target, data, value, errorMessage);
    }

    function _functionCallWithValue(address target, bytes memory data, uint256 weiValue, string memory errorMessage) private returns (bytes memory) {
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: weiValue }(data);
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}